Gaussian Quadrature

```
> restart;
```

We wish to use **Guassian quadrature** to approximate the integral $\int_{1}^{1.6} \frac{2x}{x^2-4}\, dx$.

```
> Digits:=15;
```

$$Digits := 15$$

We load the orthopoly library to get the **Legendre polynomials** needed for **Gaussian quadrature**.

```
> with(orthopoly);
```

$$[G, H, L, P, T, U]$$

It is the P that interests us. Let's look at the first six Legendre polynomials. Note that they differ from those in the text by a constant factor. Since we are only intersested in their roots, that is not a concern for our purposes.

```
> for i from 0 to 5 do
  legedre[i]:=P(i,x)
  od;
```

$$legedre_0 := 1$$

$$legedre_1 := x$$

$$legedre_2 := -\frac{1}{2} + \frac{3}{2}x^2$$

$$legedre_3 := \frac{5}{2}x^3 - \frac{3}{2}x$$

$$legedre_4 := \frac{3}{8} + \frac{35}{8}x^4 - \frac{15}{4}x^2$$

$$legedre_5 := \frac{63}{8}x^5 - \frac{35}{4}x^3 + \frac{15}{8}x$$

We choose the degree of the Legendre polynomial we wish to use.

```
> n:=5;
```

$$n := 5$$

Get the appropriate Legendre polynomial.

```
> legendre:=P(n,x);
```

$$legendre := \frac{63}{8}x^5 - \frac{35}{4}x^3 + \frac{15}{8}x$$

Find the roots of the n 'th Legendre polynomial, all of which are between -1 and 1. These are given as a table in the text on page 141.

```
> r:=[evalf(allvalues(RootOf(legendre)))];
```

$$r := [0., 0.538469310105681, 0.906179845938666, -0.538469310105681, -0.906179845938666]$$

Find the corresponding coefficients for Gaussian quadrature as given by the Theorem in the notes. These are also given on page 141.

```
> for i from 1 to n do
  c[i]:=1:
```

```
  for j from 1 to n do
  if (i<>j) then c[i]:=c[i]*(x-r[j])/(r[i]-r[j]) fi
  od:
  c[i]:=int(c[i],x=-1..1):
  od:
  printf("   r[i]                    c[i]\n");
  printf("   ----                    ----\n");
  for i from 1 to n do
  printf("%18.15f     %18.15f\n",r[i],c[i])
  od;
```

```
   r[i]                    c[i]
   ----                    ----
 0.000000000000000       0.568888888888877
 0.538469310105681       0.478628670499373
 0.906179845938666       0.236926885056187
-0.538469310105681       0.478628670499372
-0.906179845938666       0.236926885056187
```

We enter the function to integrate as a Maple function to make the change of variables easier.

```
> f:=x->2*x/(x^2-4);
```

$$f := x \rightarrow \frac{2\,x}{x^2 - 4}$$

We give the limits of integration.

```
> a:=1;b:=1.6;
```

$$a := 1$$

$$b := 1.6$$

We do the change of variables so as to integrate from -1 to 1.

```
> g:=f(((b-a)*t+(b+a))/2)*(b-a)/2;
```

$$g := \frac{0.600000000000000\,(0.300000000000000\,t + 1.30000000000000)}{(0.300000000000000\,t + 1.30000000000000)^2 - 4}$$

```
> g:=unapply(g,t);
```

$$g := t \rightarrow \frac{0.600000000000000\,(0.300000000000000\,t + 1.30000000000000)}{(0.300000000000000\,t + 1.30000000000000)^2 - 4}$$

We find the integral by Gaussian quadrature.

```
> integral:=evalf(sum('c[i]*g(r[i])','i'=1..n));
```

$$integral := -0.733968724860456$$

Compare to the value from the built-in routine.

```
> Int(f(x),x=1..8/5);
  value(%);
```

$$\int_1^{\frac{8}{5}} \frac{2\,x}{x^2 - 4}\,dx$$

$$\ln(3) + 2\ln(2) - 2\ln(5)$$

```
> exact:=evalf(%);
```

$$exact := -0.73396917508020$$

Finally, we find the absolute error.

```
> interror:=abs(integral-exact);
```

$$interror := 4.50219744\ 10^{-7}$$

NumericalAnalysis

```
> with(Student[NumericalAnalysis]);
```

[*AbsoluteError*, *AdamsBashforth*, *AdamsBashforthMoulton*, *AdamsMoulton*, *AdaptiveQuadrature*, *AddPoint*, *ApproximateExactUpperBound*, *ApproximateValue*, *BackSubstitution*, *BasisFunctions*, *Bisection*, *CubicSpline*, *DataPoints*, *Distance*, *DividedDifferenceTable*, *Draw*, *Euler*, *EulerTutor*, *ExactValue*, *FalsePosition*, *FixedPointIteration*, *ForwardSubstitution*, *Function*, *InitialValueProblem*, *InitialValueProblemTutor*, *Interpolant*, *InterpolantRemainderTerm*, *IsConvergent*, *IsMatrixShape*, *IterativeApproximate*, *IterativeFormula*, *IterativeFormulaTutor*, *LeadingPrincipalSubmatrix*, *LinearSolve*, *LinearSystem*, *MatrixConvergence*, *MatrixDecomposition*, *MatrixDecompositionTutor*, *ModifiedNewton*, *NevilleTable*, *Newton*, *NumberOfSignificantDigits*, *PolynomialInterpolation*, *Quadrature*, *RateOfConvergence*, *RelativeError*, *RemainderTerm*, *Roots*, *RungeKutta*, *Secant*, *SpectralRadius*, *Steffensen*, *Taylor*, *TaylorPolynomial*, *UpperBoundOfRemainderTerm*, *VectorLimit*]

For Gaussian quadrature, we use the Quadrature command with the option **method = gaussian[n]** where n is the number of nodes. The default is $n = 3$ if **method=gaussian** is used. Let us find the same integral as above with $n = 5$. For output, we will show **value**, **information**, and **plot**.

```
> integral:=Quadrature(f(x),x=1..1.6,method=gaussian[5],output=value)
  ;
```

$$integral := -0.733969175080205$$

```
> Quadrature(f(x),x=1..1.6,method=gaussian[5],output=information);
INTEGRAL: Int(2*x/(x^2-4),x=1..1.6) = -0.733969175
APPROXIMATION METHOD: Gaussian Quadrature
---------------------------------- INFORMATION TABLE ------------------
---------------
    Approximate Value          Absolute Error         Relative Error
        -0.733969175                    4e-15         5.450e-13 %
-----------------------------------------------------------------------
---------------
Number of Function Evaluations:     50
```

```
> Quadrature(f(x),x=1..1.6,method=gaussian[5],output=plot);
```


An Approximation of the Integral of
$f(x) = \frac{2x}{x^2 - 4}$
on the Interval [1, 1.6]
Using Gaussian Quadrature Rule with 5 nodes
Integral Value: −0.733969175080201
Approximation: −0.733969175080205
0
−0.5
−1
−1.5
−2
−2.5
1.1
1.2
1.3
1.4
1.5
1.6
x
f(x)